
Vorwort

Liebe Leserin, lieber Leser,
ich freue mich, Sie im Kreis meiner Leser begrüßen zu dürfen. Oder sind Sie bereits längst 
dabei, haben vielleicht das Excel-Praxisbuch oder eines meiner früheren Werke gekauft, in 
dem Excel zwar umfangreich und – wie ich hoffe – verständlich beschrieben ist? Aber die 
Funktionen kommen in den Büchern immer zu kurz, was auch verständlich ist bei mittler-
weile fast 450 Funktionen, die uns die Funktionspalette anbietet.
Deshalb habe ich dieses Buch geschrieben, ein Buch, das ausschließlich den Funktionen von 
Excel gewidmet ist. Natürlich werde ich Ihnen nicht einfach die Funktionen mit der Liste der 
Argumente servieren und schon gar nicht mit simplen, nichtssagenden Beispielen. Ich zeige 
Ihnen, wie ich in meiner langjährigen Praxis mit Funktionen gearbeitet habe und immer 
noch arbeite.
Denn Excel ist ein Rechenprogramm. „Kalkulation“ kommt von „kalkulieren“ und das kann 
Excel am besten, viel besser als schreiben und Kästchen malen. Zu diesem Zweck stellt 
Excel die vielen Funktionen zur Verfügung. Funktionen lassen sich von einfachen Berech-
nungen über Formelschachteln bis zu Matrixtechniken so raffiniert kombinieren, dass es 
schon fast mit Programmierung zu vergleichen ist. Aus dem elektronischen Zeilen/Spalten-
Journal, das die Ergebnisse an den Randspalten und in den Abschlusszeilen automatisch 
summieren kann, ist ein komplexes Entwicklungswerkzeug geworden, das dynamische 
Tabellenmodelle mit Verknüpfungen, Verweisen und externen Datenimports erzeugt. Wer 
diese Techniken beherrscht, macht aus Tabellen Tabellenmodelle, und wer eine Arbeits-
mappe anlegt, macht dies ja nicht, um mehr Arbeit zu haben, sondern um ein Hilfsmittel zu 
erhalten, das ihm die Arbeit abnimmt.
Lernen Sie die Funktionen. Gute Formeln schreibt nur, wer die Funktionspalette kennt und 
wie die Vokabeln der Fremdsprache gezielt einsetzen kann, um seine Aufgaben zu lösen. 
Deshalb habe ich dieses Buch geschrieben:

�� Ein Buch für Einsteiger und Eingestiegene, die so manchen Zusammenhang noch nicht 
wirklich verinnerlicht haben und sich mit gutem Halbwissen und schlechtem Gewissen 
durch den digitalen Kalkulationsdschungel kämpfen

�� Ein Buch für Vielkalkulierer, die täglich zwei- bis dreihundert Tabellen erstellen und vor 
lauter Stress keine Zeit haben, ihr meistbenutztes Werkzeug richtig zu erlernen (nach 
dem altbewährten Motto: Wie soll ich Zeit haben, meine Axt zu schärfen, ich muss doch 
Bäume fällen . . .)

�� Ein Buch für Fortgeschrittene (ein schreckliches Wort), die Excel schon gut kennen und 
viel damit machen, aber immer das Gefühl haben, das müsste doch alles noch einfacher 
oder raffinierter gehen

�� Und ein Buch für Praktiker: Ich habe in meinen Excel-Archiven gestöbert – nach 30 Jah-
ren mit Excel sammelt sich einiges an – und viele interessante Lösungen gefunden, die 
ich in Workshops oder im Kundenauftrag entwickelt habe. Zu den wichtigsten Funktio-
nen finden Sie immer ein passendes Praxisbeispiel. Diese Lösungen gebe ich gern weiter, 
sie werden Ihnen helfen, die komplexen Formeln und Funktionen besser zu verstehen.

Beispiele zum Download
Auf den Internetseiten des Verlags finden Sie alle Funktionsbeispiele und Praxislösungen 
in kompakter Form.

http://downloads.hanser.de

2010, 2013, 2016 – es geht weiter
Dieses Buch beschreibt die Funktionen in den Office- bzw. Excel-Versionen 2010, 2013 und 
2016. Mit jeder Version wurden Funktionen geändert, verbessert und neu aufgenommen. 
Ich habe die Unterschiede folgendermaßen gekennzeichnet:

Dieses Symbol kennzeichnet Abschnitte im Text, die sich auf Excel 2010 
beziehen und in Excel 2013 so nicht mehr anzutreffen sind.

Mit diesem Symbol sind Textpassagen gekennzeichnet, die Elemente, Tools 
oder Verfahren beschreiben, die in Excel 2013 neu sind.

Dieses Symbol weist auf Programmteile und Funktionen hin, die ab Excel 
2016 zur Verfügung stehen.

Hinweis:
Mit dem Update im Januar 2016 hat Microsoft noch sechs neue Funktionen in die Funkti-
onspalette aufgenommen, die ich leider bis zum Redaktionsschluss dieses Buches nicht 
mehr beschreiben konnte. Sie finden aber eine ausführliche Beschreibung im Downloadbe-
reich in der Datei Funktionen_Januar2016.doc.
Jetzt wünsche ich Ihnen viel Spaß beim Lesen, Stöbern, Lernen und Ausprobieren der For-
meln und Funktionen in Excel. Und viel Erfolg natürlich, aber der stellt sich dann von selbst 
ein . . .

Viele Grüße
Ignatz Schels

http://downloads.hanser.de

11 Logik-Funktionen

Mit Logik in Funktionen zu arbeiten, ist ein Grundprinzip der digitalen Kalkulation. Funk-
tionen wie WENN(), UND() und ODER() werden hauptsächlich in Kombinationen mit ande-
ren Funktionen verwendet, um Ergebnisse auf Grund logischer Sachverhalte zu berechnen. 
Aber auch die anderen Funktionen haben ihre Berechtigung und unterstützen die logischen 
Operationen im Funktionseinsatz.
Hier eine Übersicht über alle Logik-Funktionen:

TABELLE 11.1 Liste der Logik-Funktionen

Funktion Erklärung
FALSCH() Gibt den logischen Wert FALSCH zurück.
NICHT() Kehrt die Logik des Arguments um, das in der Formelklammer 

angegeben wird.
ODER() Gibt WAHR zurück, wenn eines der Argumente in der Klammer 

WAHR ist.
UND() Gibt WAHR zurück, wenn alle Argumente in der Klammer 

WAHR sind.
WAHR() Gibt den logischen Wert WAHR zurück.
WENN() Gibt einen durchzuführenden logischen Test an.
WENNFEHLER() Gibt den angegebenen Wert oder Text aus, wenn ein Ausdruck 

fehlerhaft ist, andernfalls den Ausdruck selbst bzw. dessen 
Ergebnis.

WENNNV() Gibt den angegebenen Wert oder Text aus, wenn das Ergebnis 
eines Ausdrucks der Fehlerwert #N/V wäre.

XODER() Gibt FALSCH zurück, wenn alle Argumente FALSCH sind, sonst 
WAHR (ausschließliches ODER).

Im Menüband finden Sie die Funktionen aus der Kategorie Logik unter dem Eintrag Logisch 
auf der Registerkarte Formeln.

BILD 11.1 Die Logik-Funktionen im Menüband

Alle Übungsdaten, Beispiele und Lösungen in diesem Kapitel finden Sie in der 
Arbeitsmappe  
Logikfunktionen Beispiele.xlsx.

■■ 11.1 Die Funktion UND()
Diese Funktion prüft die Wahrheitswerte aller Argumente und liefert als Ergebnis WAHR, 
wenn alle Argumente ein positives Ergebnis aufweisen. UND() wird meist in Verbindung mit 
anderen Funktionen zum Einsatz kommen und seltener eine eigene Formel bilden.

=UND(Wahrheitswert1;Wahrheitswert2;…)
Mit den Argumenten Wahrheitswert1;Wahrheitswert2; . . . können bis zu 30 Werte angegeben 
werden, und zwar Wahrheitswerte, logische Ausdrücke, die als Ergebnis einen Wahrheits-
wert liefern, Bezüge oder Matrizen. Enthält ein Argument leere Einträge oder Text, wird es 
ignoriert. Liefert ein Argument keinen Wahrheitswert, gibt die Funktion die Fehlermeldung 
#WERT! aus.

A1: WAHR
A2: WAHR
A3: FALSCH
A4: =UND(A1;A2)         Ergebnis: WAHR
A5: =UND(A1;A2;A3)      Ergebnis: FALSCH

TIPP: Achten Sie darauf, dass sich die Bedingungen in der UND()-Funktion 
nicht gegenseitig aufheben. Diese Formel würde beispielsweise niemals etwas 
anderes als FALSCH liefern:
=UND(A1>0;A1<0)

■■ 11.2 �Praxisbeispiel: Geringwertige 
Wirtschaftsgüter

Sie haben die Aufgabe, eine Liste mit Anschaffungen zu überprüfen und die Artikel heraus-
zurechnen, die als geringwertige Wirtschaftsgüter sofort abzuschreiben sind. Außerdem 
sollten sie einer bestimmten Abteilung zuzuordnen sein.

 �BILD 11.2  
Liste mit Anschaffungen

1.	 Schreiben Sie den Namen der Abteilung in die erste Zeile der Auswertungsspalte:

D1: Marketing

2.	 Erstellen Sie die Formel, die abprüft, ob der erste Artikel für die Abteilung gekauft und 
abgeschrieben werden kann:

D1: Marketing
D2: =UND(B2=$D$1;C2<=450)

3.	 Kopieren Sie die Formel nach unten auf die übrigen Positionen.

 �BILD 11.3  
Die UND()-Funktion prüft hier 
zwei Bedingungen ab.

11.2.1 UND() in SUMMEWENN()-Bedingungen

Die Funktion SUMMEWENN() bietet die Möglichkeit, Werte aufgrund von Bedingungen zu 
summieren:

=SUMMEWENN(Bereich;Suchkriterien;SummeBereich)

Um im obigen Beispiel die Summe der Beträge nach zwei Bedingungen zu ermitteln, müsste 
die Funktion zwei Spalten gleichzeitig durchsuchen, und das ist nicht möglich. Verwenden 
Sie stattdessen eine Matrixfunktion mit SUMME() und WENN() in Kombination. Die Be
dingungen können Sie dabei nicht immer mit UND() zusammenfassen, diese Formel würde 
beispielsweise nicht richtig rechnen:

=SUMME(WENN(UND(B2:B4=$D$1;C2:C4<=450);C2:C4))

Stellen Sie für solche Fälle die WENN-Funktionen geschachtelt in die Summe:

=SUMME(WENN(B2:B4=$D$1;WENN(C2:C4<=450;C2:C4)))

Vergessen Sie nicht, diese Matrixformel mit Strg + Umschalt + Eingabe abzuschließen.

BILD 11.4 In bedingten Summen ist eine WENN-Schachtel besser als UND().

■■ 11.3 Die Funktion ODER()
Diese Funktion prüft die Wahrheitswerte aller Argumente und liefert als Ergebnis WAHR, 
wenn eines der Argumente ein positives Ergebnis aufweist. Auch diese Funktion wird meist 
in Verbindung mit anderen Funktionen gebracht, es können bis zu 30 Argumente ange
geben werden.

=ODER(Wahrheitswert1;Wahrheitswert2; ...)
Die Argumente Wahrheitswert1;Wahrheitswert2; . . . bezeichnen Wahrheitswerte, logische 
Ausdrücke, die als Ergebnis einen Wahrheitswert liefern, Bezüge oder Matrizen. Enthält ein 
Argument leere Einträge oder Text, wird es ignoriert. Liefert ein Argument keinen Wahr-
heitswert, gibt die Funktion einen #WERT!-Fehler aus.

A1: WAHR
A2: FALSCH
A3: FALSCH
A4: =ODER(A1;A2)        Ergebnis: WAHR
A5: =ODER(A1;A2;A3)     Ergebnis: WAHR

11.3.1 Praxisbeispiel: Quartal berechnen

Ihre Tabelle enthält eine Reihe von Datumswerten. Berechnen Sie, welche dieser Werte in 
das erste Quartal (Monat 1– 3) fallen.

 �BILD 11.5  
Datumswerte in Spalte A

1.	 Schreiben Sie diese Formel:

B2: =ODER(MONAT(A2)=1;MONAT(A2)=2;MONAT(A2)=3)

2.	 Kopieren Sie die Formel auf die übrigen Zeilen.
Die Funktion ODER() gibt WAHR aus, wenn eine der Bedingungen erfüllt ist.

BILD 11.6 Mit ODER() berechnen, welches Datum ins erste Quartal fällt

11.3.2 UND() und ODER() in Matrizen

Die logischen Funktionen UND() und ODER() lassen sich auch in Matrizen sehr nützlich 
einsetzen. Hier ein einfaches Beispiel:
1.	 Schreiben Sie in den Bereich A1:B5 eine Reihe von 1er-Werten.
2.	 Schreiben Sie eine Formel mit der UND()-Funktion, die abprüft, ob alle Zellen der Matrix 

eine 1 enthalten:

=UND(A1:B5=1)

 �BILD 11.7  
UND() berechnet eine Matrix.

3.	 Drücken Sie zum Abschluss der Formel Strg + Umschalt + Eingabe, da es sich um eine 
Matrixformel handelt.

Das Ergebnis der Funktion ist FALSCH, solange nicht alle Zellen eine 1 enthalten.
Mit der ODER()-Funktion ergibt die Formel dagegen WAHR, wenn eine einzige Zelle eine 1 
enthält, auch diese Formel müssen Sie mit Strg + Umschalt + Eingabe abschließen:

=ODER(A1:B5=1)

 �BILD 11.8  
Für ein WAHR in ODER() reicht ein 
Eintrag.

11.3.3 Praxisbeispiel: Preisliste vergleichen

Sie haben die Aufgabe, die Preisliste des Gartencenters zu aktualisieren, dazu müssen Sie 
einzelne Preise berechnen und aus anderen Listen übertragen. Da bei dieser Aktion in der 
Vergangenheit immer wieder Fehler passiert sind, sichern Sie sich über eine UND()-Funk-
tion ab. Diese soll ständig überprüfen, ob die neue Preisliste noch richtig ist:

�� Alle neuen Preise müssen höher sein als die alten.
�� Kein Artikel darf in der neuen Preisliste mehr als 20 % teurer sein als in der alten Preis-

liste.

 �BILD 11.9  
Die beiden Preislisten,  
alt und neu

1.	 Markieren Sie E3:E6 und schreiben Sie diese Formel:

=B3*1,2

2.	 Drücken Sie Strg + Eingabe, um die Formel auf die markierten Zellen zu übertragen. 
Damit sind zunächst alle Preise um 20 % erhöht worden.

3.	 Schreiben Sie eine Überwachungsformel, die sicherstellt, dass die neuen Preise höher 
sind, aber nicht 120 % der alten Preise übersteigen:

D8: =UND(B3:B6<E3:E6;E3:E6<=B3:B6*1,2)

4.	 Drücken Sie zum Abschluss der Formel Strg + Umschalt + Eingabe, da es sich um eine 
Matrixformel handelt.

Das Ergebnis ist WAHR, solange die neuen Preise nicht geändert werden. Tragen Sie aber 
einen anderen Preis ein, wird die Formel FALSCH anzeigen, falls dieser eine der beiden 
Regeln verletzt:

E3: 4,80 € oder
E3: 2,20 €

11.3.3.1 Variable Listen oder Tabellen
Wenn Sie mit variablen Listen arbeiten, deren Länge sich ab und zu ändert, weisen Sie den 
beiden Listen je einen Bereichsnamen zu (z. B. PLISTE_ALT für A3:B6 und PLISTE_NEU für 
D3:E6). Vergleichen Sie dann in der Matrixformel, ob die zweite Spalte noch die Regeln 
einhält:

=UND(INDEX(Pliste_Alt;;2)<INDEX(PListe_Neu;;2);INDEX(PListe_Neu;;2)<=INDEX 
(Pliste_Alt;;2)*1,2)

Noch eleganter lässt sich die Aufgabe mit Tabellen lösen. Weisen Sie beide Preislisten als 
Tabelle aus (Einfügen/Tabelle) und tragen Sie Spaltenüberschriften ein. Die Formel mit 
strukturierten Verweisen vergleicht dann einfach die beiden Spalten, als Argumente werden 
die Spaltenbezeichnungen verwendet. Hier zum Beispiel für die Tabellen tbl_PreislisteAlt 
und tbl_PreislisteNeu mit der Spalte Preis:

=UND(tbl_PreislisteAlt[Preis]<tbl_PreislisteNeu[Preis];tbl_PreislisteNeu 
[Preis]<=tbl_PreislisteAlt[Preis]*1,2)

BILD 11.10 Die Formel prüft zwei Tabellenspalten ab.

■■ 11.4 Die Funktion NICHT()
Mit dieser Funktion wird ein Ergebnis negiert oder umgekehrt. Das Ergebnis der Funktion 
ist FALSCH, wenn der Wert WAHR als Ergebnis bringen würde, und WAHR, wenn der Argu-
mentwert das Ergebnis FALSCH hat.

=NICHT(Wahrheitswert)
Das Argument Wahrheitswert enthält den Wert WAHR oder FALSCH, der wahlweise mit oder 
ohne Anführungszeichen geschrieben wird, oder einen Bezug auf einen Wahrheitswert, 
einen Bereichsnamen oder eine Matrix, die einen Wahrheitswert zum Ergebnis hat.

A1: WAHR
A2: "FALSCH"
A3: =NICHT(A1)          Ergebnis: FALSCH
A4: =NICHT(A2)          Ergebnis: WAHR

Wie die meisten Logik-Funktionen wird auch NICHT() fast immer in Verbindung mit ande-
ren Funktionen eingesetzt. Diese Funktion vereinfacht die Formelschreibung, wenn in einer 
Bedingung ein längerer Ausdruck benutzt wird und zum Beispiel nur der positive Ergebnis-
wert zählt.
Hier im Beispiel werden alle Werte im Bereich A1:A9 summiert, die nicht 1 sind. Achten Sie 
darauf, dass die Formel mit Strg + Umschalt + Eingabe als Matrixformel abzuschließen ist:

 �BILD 11.11  
NICHT() negiert das Ergebnis 
oder die Bedingung.

■■ 11.5 Die Funktion WENN()
Diese Funktion dürfte neben der SUMME() zu den häufigsten Funktionen zählen, die zum 
Einsatz kommen. WENN() wird immer dann genommen, wenn das Ergebnis von mehr als 
einer möglichen Bedingung abhängt.

=WENN(Prüfung;Dann_Wert;Sonst_Wert)
Mit dem Argument Prüfung geben Sie eine Bedingung oder einen Wert an. Das Argument 
Dann_Wert liefert das Ergebnis, wenn diese Bedingung logisch WAHR ist oder als Wahr-
heitswert WAHR gewertet werden kann. Ist das Ergebnis der Bedingung FALSCH oder wird 

es von Excel mit dem Ergebnis FALSCH gewertet, liefert die Funktion das zweite Argument 
Sonst_Wert als Ergebnis.
Dann_Wert und Sonst_Wert sind jeweils optional, sie müssen nicht angegeben werden. Das 
Ergebnis der Funktion ist in diesem Fall der Wahrheitswert WAHR oder FALSCH.

A1: 100
A2: =WENN(A1=100;)        Ergebnis: WAHR
A2: =WENN(A1>100;"OK")    FALSCH

Verwenden Sie zur Formulierung einer Bedingung im ersten Argument einen dieser logi-
schen Operatoren:

=     (gleich)
>=    (größer oder gleich)
<=    (kleiner oder gleich)
<>    (ungleich)
A1: 120
A2: =WENN(A1>100;"größer 100";"kleiner oder gleich 100")

Das erste Argument kann auch ein Text sein, der abgeprüft wird, in diesem Fall mit An
führungszeichen:

A1: "Sommer"
A3: =WENN(A1="Sommer";"Biergarten!";"Schifoan!")

Ausdrücke, die die Funktion als Wahrheitswerte interpretieren kann, müssen gar nicht als 
Bedingung formuliert werden:

A1: 1
A2: WAHR
A3: =WENN(A1;"negativ";"positiv")        Ergebnis: negativ
A4: =WENN(A2;"Richtig";"Nicht richtig")  Ergebnis: Richtig

11.5.1 Praxisbeispiel: Postleitzahlen sortieren

Ihre Adressenliste enthält u. a. die Postleitzahlen von Adressen. Das Länderkennzeichen 
steht an erster Stelle, sodass Sie leicht zwischen deutschen und österreichischen Kontakten 
unterscheiden können.

 �BILD 11.12  
PLZ-Liste mit Länderkennzeichen

1.	 Schreiben Sie eine WENN-Funktion, die das Land für die Adresse in einer zweiten Spalte 
ausgibt. Damit nur das erste Zeichen der Postleitzahl abgeprüft wird, benutzen Sie die 
Funktion LINKS():

B1: Land
B2: =WENN(LINKS(A2;1)="D";"Deutschland";"Österreich")

2.	 Kopieren Sie die Formel per Doppelklick auf das Füllkästchen auf die übrigen Zeilen der 
Liste.

BILD 11.13 Mit WENN() wird das erste Zeichen der PLZ abgeprüft.

11.5.2 WENN() geschachtelt

Was tun, wenn mehr als eine Bedingung zu formulieren ist? Das Beispiel mit der Post
leitzahlenliste drängt diese Frage förmlich auf: Wie sieht die Funktion aus, wenn auch an-
dere und vor allem mehr als zwei Länder in der Liste vertreten sind?
In diesem Fall schachteln Sie die WENN()-Funktion. Dazu wird anstelle des Sonst-Werts eine 
weitere WENN-Funktion gestartet, und zwar ohne =-Zeichen. Vergessen Sie nicht, die For-
mel mit zwei Klammern abzuschließen:

=WENN(Bedingung1;Dann_Wert1;WENN(Bedingung2;Dann_Wert2;Sonst_Wert))
Ist die in Bedingung1 formulierte Aussage WAHR, wird Dann_Wert1 das Ergebnis liefern. Ist 
sie nicht wahr, startet die zweite WENN-Funktion mit Bedingung2. Ist diese WAHR, wird 
Dann_Wert2 das Ergebnis liefern. Sind beide Bedingungen FALSCH, holt sich die Funktion 
das Ergebnis aus dem Argument Sonst_Wert.

11.5.2.1 Praxisbeispiel: Teilnehmerliste nach Alter gruppieren
Die Teilnehmerliste für den Stadtmarathon liegt vor. Sie haben die Aufgabe, die angemelde-
ten Läufer in Altersgruppen einzuteilen:

bis 19 Jahre: Jugend
20 bis 29 Jahre: Junioren
30 bis 49 Jahre: Herren
ab 50 Jahre: Senioren



11.5 Die Funktion WENN() 479

 

11

 �BILD 11.15  
Teilnehmerliste mit Alter – berechnen Sie die Altersgruppe.

1.	 Schreiben Sie die Formel, die über eine geschachtelte WENN-Funktion berechnet, in 
welcher Altersgruppe der erste Läufer startet:

C1: Altersgruppe
C2: =WENN(B2<20;"Jugend";WENN(B2<30;"Junioren";WENN(B2<50;"Herren"; 

"Senioren")))

2.	 Kopieren Sie die Formel nach unten auf die übrigen Zeilen.

BILD 11.16 Die Teilnehmer sind in Altersgruppen eingeteilt.

In diesem Beispiel sind die Bedingungen einfach zu formulieren, weil WENN() diese der 
Reihe nach (von links) abfragt. Achten Sie darauf, dass sich Bedingungen nicht versehent-
lich ausschließen. Verwenden Sie im Zweifelsfall UND() und ODER(), um mehr als eine Be-
dingung zu formulieren. Hier zum Beispiel eine Bedingung, die eindeutig die Altersgruppe 
zwischen 30 und 50 Jahren bezeichnet:

=WENN(UND(B2>=30;B2<=50);…)

11.5.2.2 Mehr als sieben WENN()-Funktionen schachteln
Sie können bis zu sieben WENN-Funktionen ineinanderschachteln, mehr sind nicht mög-
lich. Es gibt aber einen Trick, wie Sie diese Grenze umgehen können.
Schreiben Sie die WENN-Funktionen mit einem &-Zeichen als Textkette, formulieren Sie nur 
das positive Ergebnis und stellen Sie das Argument Sonst_Wert mit zwei Anführungszeichen 
als leere Zeichenkette ein:

=WENN(Bedingung1;Dann_Wert;““)& WENN(Bedingung2;Dann_Wert2;““) …& 
WENN(Bedingungn;Dann_Wertn;““)

Mit dieser Technik können Sie theoretisch unendlich viele Bedingungen schachteln, nur 
durch den Zellinhalt und den von Excel nutzbaren Hauptspeicher ist ein physikalisches 
Limit gesetzt. Hier ein Beispiel: Die Spalte A enthält verschiedene Ländernamen, mehr als 
zehn. Es gilt die im jeweiligen Land gesprochene Sprache zu „berechnen“.

 �BILD 11.16  
Berechnen Sie die Landessprachen.

Eine geschachtelte WENN-Funktion kommt nicht infrage, weil diese nur sieben Alternativen 
zulässt. Sie können aber einzelne WENN-Bedingungen formulieren und diese mit der &-Ver-
kettung aneinanderreihen. Jede Funktion, deren Bedingung nicht WAHR ist, wird als leere 
Zeichenkette interpretiert, und übrig bleibt das Ergebnis der Bedingung, die als einzige 
positiv ist:

B1: Landessprache
B2: =WENN(A2="Italien";"Italienisch";"")&WENN(A2="Spanien";"Spanisch";"")& 

WENN(A2="Frankreich";"Französisch";"")&WENN(A2="Niederlande"; 
"Holländisch";"")&WENN(A2="Belgien";"Belgisch";"")&WENN(A2="Polen"; 
"Polnisch";"")&WENN(A2="Großbritannien";"Englisch";"")&WENN(A2="Irland";
"Irisch";"")&WENN(ODER(A2="Österreich";A2="Schweiz";A2="Deutschland"); 
"Deutsch";"")

Bevor Sie solche Formelmonster erstellen, prüfen Sie, ob nicht eine alternative Funktion 
besser wäre. Für das Beispiel mit der Landessprache könnten Sie auch eine Liste anlegen, 
in der Land und Sprache in zwei Spalten nebeneinanderstehen:

E2: Italien
F2: Italienisch
E3: Spanien
F3: Spanisch

Mit der Funktion SVERWEIS() lässt sich einfach abprüfen, in welcher Zeile der Liste sich das 
gesuchte Land befindet:

B2: =SVERWEIS(A2;$E$2:$F$12;2;FALSCH)

Sehen Sie sich auch die Funktion WAHL() an. Wenn die Bedingung einen numerischen In-
dex formulieren kann, geben Sie diesen in einer WAHL-Funktion an. Typisches Beispiel: 
Eine Monatsreihe von Januar bis Dezember wird nach der Monatszahl durchsucht:



11.6 Die Funktion WENNFEHLER() 481

 

11

A1: 3
A2: =WAHL(A1;“Januar“;“Februar“;“März“;“April“;“Mai“;“Juni“;“Juli“;“August“;

“September“;“Oktober“;“November“;“Dezember“)

BILD 11.17 Mehr als sieben WENN() mit &-Verknüpfung

■■ 11.6 Die Funktion WENNFEHLER()
Mit WENNFEHLER() werden Fehler in Formeln aufgedeckt oder – in den meisten Fällen – 
absichtliche Fehler verdeckt:

=WENNFEHLER(Wert;Wert_falls_Fehler)
Das Argument Wert ist der Ausdruck oder die Formel, die auf einen Fehler überprüft wird. 
Der Wert kann folgende Fehlerwerte zum Ergebnis haben:

TABELLE 11.2 Fehlerwerte der Funktion WENNFEHLER()

#NV 
#WERT! 
#BEZUG! 
#DIV/0!

#ZAHL! 
#NAME? 
#NULL!

Im zweiten Argument Wert_falls_Fehler wird ein Text oder eine Formel angegeben, die als 
Ergebnis angezeigt wird, wenn der erste Ausdruck einen Fehler produziert. Beide Argu-
mente müssen angegeben werden:

=WENNFEHLER(100/0;"")          Ergebnis: leere Zelle
=WENNFEHLER(100/0;"Fehler")    Ergebnis: „Fehler“
=WENNFEHLER(100/10)            Ergebnis: 10

Bis Excel 2003 mussten Fehler etwas umständlicher abgefangen werden, wie dieses Beispiel 
zeigt. Die Liste enthält statistische Angaben über deutsche Bundesländer, über eine Formel 
mit der Funktion SVERWEIS wird die Information abgerufen:

BILD 11.18 Statistische Informationen und eine Auswertung über SVERWEIS()

Um das Ergebnis abzusichern, kann der Verweis mit den beiden Funktionen WENN() und 
ISTFEHLER() kombiniert werden:

=WENN(ISTFEHLER(SVERWEIS(F2;A1:C17;3;FALSCH));"";SVERWEIS(F2;A1:C17;3;FALS
CH))

Die Funktion WENNFEHLER() erledigt diese Aufgabe etwas eleganter, hier muss die Ver-
weisfunktion nur einmal eingeben werden:

=WENNFEHLER(SVERWEIS(F2;A1:C17;3;FALSCH);"")

Werden in WENNFEHLER() Matrizen (Arrays) benutzt, berechnet die Funktion jedes ein-
zelne Element der Matrix und gibt für dieses einen Fehler aus. Sie können damit ganze 
Zeilen oder Spalten von Bereichen berechnen und brauchen nur eine Formel dazu. Hier im

 �BILD 11.19  
Kostendifferenzen berechnen



11.6 Die Funktion WENNFEHLER() 483

 

11

Beispiel eine Gegenüberstellung von Kosten einzelner Kostenstellen. Die Formel berechnet 
die Differenz der beiden Jahre und stößt natürlich auf einen Fehler, wenn für das zweite Jahr 
noch keine Angabe (k. A.) gemacht wurde:
1.	 Markieren Sie den Bereich D5:D11 und geben Sie die Formel ein:

=WENNFEHLER($C$5:$C$11-$B$5:$B$11;““)

2.	 Drücken Sie Strg + Eingabe, um die Formel auf alle markierten Zellen zu verteilen.
Eine Matrixformel, mit Strg + Umschalt + Eingabe erzeugt, ist zwar auch möglich, aber 
nicht nötig, da WENNFEHLER() die Elemente einzeln berechnet. Wenn Sie mit variablen 
Bereichen arbeiten, die ihre Größe ändern, weisen Sie dem Bereich einen Bereichsnamen zu 
und verwenden die INDEX()-Funktion, um einzelne Spalten zu adressieren:

=WENNFEHLER(INDEX(KSLISTE;;3)-INDEX(KSLISTE;;2);"")

11.6.1 Mit WENNFEHLER() in Tabellen rechnen
Die Tabelle ist für variable Bereiche das ideale Medium, sie erweitert sich automatisch, 
wenn neue Zeilen oder Spalten hinzukommen, und bietet die Möglichkeit, die Berechnun-
gen zu integrieren:
1.	 Markieren Sie den Bereich mit den Kostenstellennummern und den Ausgaben der ein

zelnen Jahre.
2.	 Erstellen Sie mit Einfügen/Tabellen/Tabelle eine Tabelle, weisen Sie dieser über Tabellen-

tools/Entwurf/Eigenschaften den Tabellennamen tbl_Ksliste zu.
3.	 Ziehen Sie die Markierung rechts unten an der letzten Zelle nach rechts, um eine neue 

Spalte einzufügen. Tragen Sie als Spaltenüberschrift Differenz ein.
4.	 Schreiben Sie in die erste Zeile dieser Spalte (nach der Überschrift) diese Formel. Für die 

Spaltennamen in eckigen Klammern ziehen Sie die Markierung über alle Zeilen der 
Spalte:

=WENNFEHLER([Ausgaben 2013]-[Ausgaben 2014];"")

Die Formel wird automatisch auf alle Zeilen der Tabelle kopiert und selbstständig nach
kopiert, wenn neue Zeilen hinzugefügt werden.

BILD 11.20 Mit WENNFEHLER wird das Arbeiten in Tabellen einfacher.

■■ 11.7 Die Funktion WENNNV()
Mit Excel 2013 hat die Funktionspalette noch eine weitere Sicherungsfunktion bekommen. 
WENNNV() gibt den im zweiten Argument angegeben bzw. berechneten Wert oder Text aus, 
wenn der Ausdruck im ersten Argument zu einem Fehler führt:

=WENNNV(Wert;Wert_bei_NV)
Wert ist das Argument, das einen Wahrheitswert liefert. Das kann eine Formel oder Funk-
tion sein, die zu einem Ergebnis führt, oder einfach nur eine Zahl.
Wert_bei_NV wird in der Formelzelle ausgegeben, wenn das erste Argument den Fehlerwert 
FALSCH liefert.

11.7.1 Praxisbeispiel: Kostenstelle suchen

1.	 Erstellen Sie eine Tabelle mit Kostenstellennummern und jährlichen Ausgaben. Geben 
Sie der Tabelle über die Tabellentools den Namen tbl_Ausgaben.

2.	 Schreiben Sie die gesuchte Kostenstelle in eine Zelle:

F5: Kostenstelle:
G5: 300-80

3.	 Berechnen Sie mit einem SVERWEIS() auf die Kostenstellennummer die Ausgaben des 
aktuellen Jahres (Spalte 2):

F6: Ausgaben 2014:
G6: =SVERWEIS(G5;tbl_Ausgaben;2;FALSCH)

Das Ergebnis ist die Fehlerfunktion #N/V, da die angegebene Kostenstelle nicht in der ers-
ten Spalte der Tabelle enthalten ist. Schachteln Sie die Formel mit einer WENNNV()-Funk-
tion und geben Sie einen passenden Text aus:

G6: =WENNNV(SVERWEIS(G5;tbl_Ausgaben;2;FALSCH);"Kostenstelle nicht vorhan-
den!")

BILD 11.21 SVERWEIS-Suche absichern mit WENNV()

■■ 11.8 Die Funktion XODER()
Liefert den Wahrheitswert FALSCH, wenn eine der angegebenen Bedingungen FALSCH ist. 
Sind alle Bedingungen in den Argumenten WAHR, liefert die Funktion den Wahrheitswert 
WAHR.

=XODER(Argument1;Argument2; … Argumentn)
Argument1 bezeichnet das erste Argument, das ein logischer Ausdruck, eine Formel oder 
nur ein Zahlenwert sein kann. Es können bis zu 254 Argumente angegeben werden.

11.8.1 Praxisbeispiel: Budgetüberziehung

Die Liste enthält monatliche Ausgaben und zu jedem Monat ein festgelegtes Budget. Um 
festzustellen, ob in einem der Monate das Budget überzogen wurde, berechnen Sie, ob alle 
Ausgaben in Zeile 7 kleiner oder gleich dem Budget sind:

 �BILD 11.22  
Budgets und Ausgaben

J6: Kosten >=Budget:
J7: XODER(B7:G7<=B6:G6)

Da es sich bei dieser Formel um eine Matrixformel handelt, müssen Sie zum Abschluss (und 
nach jeder Änderung) die Tastenkombination Strg + Umschalt + Eingabe drücken. Ist das 
Ergebnis FALSCH, gibt es keine Budgetüberziehungen, bei WAHR ist mindestens ein Wert 
in Zeile 7 größer als in Zeile 6.
Schachteln Sie die Formel in eine WENN-Funktion, können Sie der Zelle noch einen aus
sagekräftigen Text mitgeben. Auch hier müssen Sie aber zum Abschluss Strg + Umschalt + 
Eingabe drücken.

=WENN(XODER(B7:G7<=B6:G6)=FALSCH;"Keine Budgetüberziehungen"; 
"Budgetüberziehung")

BILD 11.23 Eleganter als WAHR und FALSCH: WENN-Funktion mit Text

